


AD-A232 812

AD-A232 812

ARE TM(UHR)90313

NOVEMBER 1990

COPY No 33



DTIC FILE COPY

GENERATING SHIP MOTION TIME HISTORIES

DTIC
ELECTE
MAR 27 1991

P Crossland

DISTRIBUTION STATEMENT A

Approved for public release;
Distribution Unlimited

This document is the property of Her Majesty's Government and Crown copyright is reserved. Requests for permission to publish its contents outside official circles should be addressed to the Issuing Authority.

Admiralty Research Establishment
Haslar Gosport Hants PO12 2AG

© British Crown Copyright 1990/MOD

Published with the permission of the
Controller of Her Britannic Majesty's
Stationery Office

UNLIMITED

91 3 20 018

0089520

CONDITIONS OF RELEASE

BR-116007

DRIC U

COPYRIGHT (c)
1988
CONTROLLER
HMSO LONDON

DRIC Y

Reports quoted are not necessarily available to members of the public or to commercial organisations.

UNLIMITED

ARE TM(UHR)90313

November 1990

GENERATING SHIP MOTION TIME HISTORIES

By

P Crossland

Summary

PAT-86 is a suite of seakeeping computer programs which may be used for predicting motion spectra of ships in waves. This Technical Memorandum describes a method of generating time histories from these motion spectra. A computer program which generates time histories has been written and is listed in the Appendix.

Comparisons of the motion spectra from PAT-86 are made with the spectra of the generated time histories. These comparisons are shown to be satisfactory.

Admiralty Research Establishment
Haslar Gosport Hants PO12 2AG

© British Crown Copyright 1990/MOD

Published with the permission of the
Controller of Her Britannic Majesty's
Stationery Office

Accession For	
NTIS CEA&I	✓
DIC TAB	
Unannounced	
Justification	
By	
Distribution	
Availability	
Dist	
A-1	

UNLIMITED

Contents

	Page
Notation.	iv
1. Objectives.	1
2. Introduction.	1
3. PAT-86 Output.	1
4. Waves.	2
5. Rigid Body Motions.	3
6. Total Motions.	3
6.1. Generation using PAT-86 Total Motion Spectra.	3
6.2. Generation using Rigid Body Motion Time Histories.	4
7. Examples.	4
8. Results.	4
9. Conclusion.	5
References.	6
Appendix A. Fortran Program.	7
Figure 1. Spectra.	
Figure 2. Time Histories.	
Figure 3. Variation of Heave-Pitch Correlation Coefficient with Heading.	
Figure 4. Total Motion Spectra.	
Figure 5. Total Motion Time History.	

Notation

$\delta\omega$	Frequency interval	[Rad/s]
$\delta\omega_e$	Encounter frequency interval	[Rad/s]
ε_s	Total motion phase	[Rad]
ε_ζ	Wave phase	[Rad]
ε_z	Heave phase	[Rad]
g	Acceleration due to gravity	[m/s ²]
μ	Heading	[degs]
N	Number of frequency intervals	
s_{no}	Total motion amplitude	[m]
S_θ	Pitch spectrum	[deg ² /(Rad/s)]
S_ζ	Wave spectrum	[m ² /(Rad/s)]
S_z	Total motion spectrum	[m ² /(Rad/s)]
$s(t)$	Total motion time history	[m]
ζ_{no}	Wave amplitude	[m]
$\zeta(t)$	Wave time history	[m]
t	time	[s]
T_H	Signal repeat period	[s]
$\theta(t)$	Pitch time history	[degs]
U	Speed	[m/s]
ω	Wave frequency	[Rad/s]
ω_e	Encounter wave frequency	[Rad/s]
x_{B1}	Distance from LCG to point of total motion calculation	[m]
z_{no}	Heave amplitude	[m]
$z(t)$	Heave time history	[m]

GENERATING SHIP MOTION TIME HISTORIES

by P Crossland

1. OBJECTIVES

There has been a recent demand for time histories of ship motions to be calculated for a given ship in a given sea condition. The PAT-86 suite of ship motion computer programs calculates motion spectra rather than the instantaneous values or time history.

This technical memorandum outlines a method of generating time histories from the five principle ship motion spectra and from the total motion spectra. The problems associated with generating multiple time histories are related to the choice of phases. Random phases must be chosen in such a way that the correct correlation between the motions is maintained.

Examples of time histories are shown and, as one check, the motion time history at a point is calculated from the heave and pitch motion and compared with that generated from a total motion spectrum.

2. INTRODUCTION

As the ship passes through the waves it responds in six rigid body modes, surge, sway, heave, roll, pitch and yaw. The sign convention and units are given in Reference 1. These rigid body motions are superimposed on one another leading to a compound motion at any point on the ship.

These rigid body motions and compound motions are calculated by the PAT-86 suite of computer programs at ARE (Haslar). The suite, first developed around 1980, was based on state-of-the-art theory. The prediction of ship motion is calculated from frequency domain dynamic analysis of the ship in waves and the process is divided into three stages.

- a. Calculation of motion in regular (sinusoidal) waves over a range of frequencies of interest. This calculation uses strip theory.
- b. Combination of these regular wave responses, or transfer functions, with a wave spectrum to give a response spectrum.
- c. Integration of the response spectrum to give variance of motion displacement. Calculation of the second and fourth moments will give variances of velocity and acceleration respectively.

A full description of the theory is given in Reference 2. This TM outlines a method of converting the response spectrum into time histories of that response. The method used is summation of sine waves with randomly generated phases.

3. PAT-86 OUTPUT

The output from the PAT-86 suite contains the regular wave calculation (transfer function and phase) and the irregular wave calculation of the 5 rigid body motions and total motions at specified points. (Surge is neglected). All this information is required if time histories are to be

generated. So it must be extracted from the main PAT-86 output file by some suitable means.

4. WAVES

A wave spectrum is input into PAT-86 and the resulting motion spectra are calculated. The wave spectrum is in the wave frequency domain and the motion spectra are in the encounter frequency domain.

a. Warping

The first step in generating time histories is to change the wave spectrum to a spectrum in the wave encounter frequency domain. This process is known as warping.

The wave encounter frequency becomes:

$$\omega_e = \omega - \frac{\omega^2 U \cos(\mu)}{g} \quad (1)$$

The spectral ordinate becomes:

$$S_\zeta(\omega_e) = \frac{S_\zeta(\omega)}{\left(1 - \frac{2\omega U \cos(\mu)}{g}\right)} \quad (2)$$

b. Amplitude Spectra

The Power Spectra output from PAT-86 need to be converted to amplitude spectra using the following formula for the wave amplitude associated with the n^{th} frequency:

$$\zeta_{n0} = \sqrt{2 S_\zeta(\omega_e) \delta \omega_e} \quad (3)$$

where

$$\delta \omega_e = \left(1 - \frac{2\omega U \cos(\mu)}{g}\right) \delta \omega \quad (4)$$

c. Generation

The time history is generated by adding together a large number of component sine waves whose amplitude is given by the amplitude spectrum. Thus, the time history is given by:

$$\zeta(t) = \sum_{n=1}^N \zeta_{n0} \cos(\omega_e t + \epsilon_\zeta) \quad (5)$$

for $0 \leq t \leq T_H$

where T_H is the signal repeat period given by

$$T_H = \frac{2\pi}{\delta\omega} \quad (6)$$

and ϵ_ζ is a randomly chosen phase lying between $-\pi$ and π (different for each frequency).

5. RIGID BODY MOTIONS

PAT-86 outputs the rigid body motions with phase relative to a wave peak at the CG of the ship. This calculation is carried out in regular waves. The method of synthesising time histories of the rigid body motions is similar to that described in Section 4. However the regular wave phases are added to the random phases generated in 4.

Thus, taking heave as an example.

The heave amplitude associated with the n^{th} frequency is given by

$$z_{n0} = \sqrt{2S_z(\omega_e)\delta\omega_e} \quad (7)$$

and the time history is given by

$$z(t) = \sum_{n=1}^N z_{n0} \cos(\omega_e t + (\epsilon_\zeta + \epsilon_z)) \quad (8)$$

for $0 \leq t \leq T_H$

ϵ_z is the heave phase taken from the regular wave response (transfer function). The other 4 motions (sway, roll, pitch and yaw) are dealt with in a similar way.

6. TOTAL MOTIONS

6.1. Generation using PAT-86 Total Motion Spectra

PAT-86 outputs the total motions (absolute vertical, lateral and relative vertical) at specified points. Again the phases ϵ_z are found in regular waves and these must be combined with the random phases ϵ_ζ found in 4, to generate a time history from the calculated motion spectrum. The motion amplitude spectrum is found from

$$s_{n0} = \sqrt{2S_s(\omega_e)\delta\omega_e} \quad (9)$$

and the resulting time history is given by

$$s(t) = \sum_{n=1}^N s_{n0} \cos(\omega_e t + (\epsilon_\zeta + \epsilon_s)) \quad (10)$$

6.2. Generation using Rigid Body Motion Time Histories

An alternative method of calculating total motion time histories is now described. Generate the time histories of the five rigid body motions using the method outlined above in Section 5. Then, combine the time histories to obtain the total motions at a point.

Take the simple case of vertical motion at a point x_{B1} metres forward of the LCG and on the ship centreline. The time history of vertical motion is given by

$$s(t) = z(t) - x_{B1}\theta(t) \quad (11)$$

where the pitch angle, θ , is in radians.

7. EXAMPLES

A ship motion calculation was carried out using the PAT-86 suite. The calculation considered only the vertical plane motions of a ship travelling in head seas.

The program calculated the absolute vertical motion at the FP on the centreline of the ship. Time histories of heave and pitch were calculated and the resulting total motion time history was determined using equation (11). As a check, a total motion time history was generated from the motion spectrum using equation (10). Comparisons were made of both the total motion time histories and their spectra.

A program was written to generate the time histories and carry out the comparisons. The program is shown in Appendix A.

As a further study, correlation coefficients of the heave and pitch time histories were calculated in accordance with the method described in Reference 3.

8. RESULTS

Figure 1 shows the wave, heave and pitch spectra, found in PAT-86, from which the time histories were generated. Also shown are the spectra calculated from the generated time histories to confirm that they are correct. The small differences between the two spectra are probably due to the short signal repeat period used. Time histories from the spectra are shown in Figure 2. A 50 second section of each time history is shown, but about 1200 seconds was generated in all. Figure 3 shows the variation of Heave-Pitch correlation coefficient with heading. The curve calculated from the generated time histories is compared with the correlation curve found from the seakeeping trial (Reference 3). The trends shown in both curves are similar, however the actual values are quite different. This is not particularly surprising since two different ships are being compared, but it illustrates that the time histories are correctly correlated. This would suggest that the phase information has been chosen correctly.

Figure 4 shows total motion spectra at the FP. The total motion spectrum as found in PAT-86 is compared with the total motion spectrum calculated from the heave and pitch time histories. The small differences in the two spectra may again be attributed to the complete time history being two

short. Finally, the total motion time history generated from the total motion spectrum is shown in Figure 5. The total motion time history found from heave and pitch is also shown but it is actually coincident with the first time history.

9. CONCLUSION

This Technical Memorandum describes a method of generating time histories. The comparisons shown indicate that the choice of phasing the random signal is correct.

REFERENCES

1. P R Loader, R N Andrew. User Guide for the PAT-86 Suite of Ship Motion Computer Programs. ARE TM(UHR) 86301, January 1986. UK UNCLASSIFIED.
2. A R J M Lloyd. Seakeeping: Ship Behaviour in Rough Weather. Ellis Horwood Series in Marine Technology 1989.
3. A R J M Lloyd, A Green. The Joint Probabilities of Occurrence of Ship Motions. ARE TM(UHR) 86317, December 1986. UK UNCLASSIFIED.

Appendix A

FORTRAN PROGRAM

PROGRAM HISTORY

```

*
* Program to generate time history given a spectrum.
* The phase information is taken from a uniform random
* variable between -PI AND PI
*
*
*           Written by
*           P Crossland  04-Oct-1990
*
*
      REAL*8 X,G05CAF
      REAL  XG(3200),A(50),B(50),SEC(50),W(50),S(50),ZETA(50)
*          ,E(50),HEAD(2),WE(50),SE(50),XGS(3200),XGC(3200)
*          ,TIME(3200),DWE(50),SC(50),EZ(50),ETHETA(50)
*          ,SZ(50),STHETA(50),ES(50)
*          ,SS(50),C(3200),TOT(50),HEAV(50),PITC(50),Z(3200)
*          ,THETA(3200),FP(3200)
      COMMON/DATA/NTIME,WE,TIME,DWE,ANS
*
* Data statement containing necessary variables which must be
* changed according to the PAT-86 output.
*
      DATA OMEGA1,OMEGAF,DW,U GRAV /0.05,2.50,0.05,10.296,9.81/
      DATA SUM4/5*0.0/
      DATA HEAD /45.0,180.0/
*
* The information contained in the above data statements can be
* obtained from the output file from PAT-86
*
*
* Open input file WAVE.0 containing a maximum of 50 spectral
* ordinates with frequency as the independent variable.
*
      OPEN(UNIT=5,FILE='WAVE.0')
*
* Open rest of the files
*
*       REG.0      - contains the phase information for
*                   pitch and heave
*       IRREG.0    - contains the heave and pitch spectra
*       TREG.0     - contains the phase information for motions
*       TIRREG.0   - contains the total motion spectrum
*
      OPEN(UNIT=7,FILE='REG.0')
      OPEN(UNIT=9,FILE='IRREG.0')
      OPEN(UNIT=10,FILE='TREG.0')
      OPEN(UNIT=11,FILE='TIRREG.0')
*
* Open output files SPECTRUM and TIME, the calculated spectra
* and the calculated time histories

```

```

*
      OPEN(UNIT=8,FILE='TIME')
      OPEN(UNIT=6,FILE='SPECTRUM')
*
* Initialise variables and declare constants.
*
      PRINT*, 'What is the heading in degrees'
      READ(*,*) HEAD(2)
      PI = 4.0*ATAN(1.0)
      NPOINTS = 0
      IWT = 0
      IFAIL = 0
*
*
* Read in the frequencies and spectral ordinates and phase
* information
* D1,D2,D3,D4,ID are insignificant information.
*
      DO 10 I=1,50
        READ(5,*,END=11) W(I),S(I)
        READ(7,*,END=11) D1,D2,D3,EZ(I),D4,ETHETA(I)
        READ(9,*,END=11) ID,D2,SZ(I),STHETA(I)
        READ(10,*,END=11) D1,D2,D3,ES(I)
        READ(11,*,END=11) ID,D2,SS(I),D3,D4
*
* Change phases into radians and count the number of frequency
* steps.
*
        EZ(I) = EZ(I)*PI/180.0
        ES(I) = ES(I)*PI/180.0
        ETHETA(I) = ETHETA(I)*PI/180.0
        NPOINTS = NPOINTS + 1
10      CONTINUE
11      CONTINUE
*
* Warp the wave spectrum to encounter frequencies
* Change Heading input to radians
*
      HEAD(1) = HEAD(1)/180.0*PI
      HEAD(2) = HEAD(2)/180.0*PI
      DO 100 I = 1,NPOINTS
        WE(I) = W(I) - W(I)*W(I)*U*COS(HEAD(2))/GRAV
        SE(I) = S(I)*GRAV/(GRAV-2.0*W(I)*U*COS(HEAD(2)))
100     CONTINUE

      C1 = 2.0*U*COS(HEAD(2))/GRAV
      C2 = -2.0*U/GRAV
*
* Calculate the amplitude spectrum for the required range
* of frequencies.
*
      DO 12 I = 1,NPOINTS
        DWE(I) = (1.0-C1*W(I))*DW
        TOT(I) = SQRT(2.0*SS(I)*DWE(I))
        HEAV(I) = SQRT(2.0*SZ(I)*DWE(I))

```

```

        PITC(I) = SQRT(2.0*STHETA(I)*DWE(I))
        ZETA(I) = SQRT(2.0*SE(I)*DWE(I))
12    CONTINUE
*
* Calculate the repeat period of the signal to be generated
*
        TH = 2.0*PI/DW
*
* Generate the random phases using NAG libraries and infer the
* phases for heave,pitch and total motion
*

        CALL G05CBF(0)
        DO 13 I = 1,NPOINTS
            X = G05CAF(X)
            E(I) = (FLOAT(X)-0.5)*PI*2.0
            EZ(I) = E(I) + EZ(I)
            ETHETA(I) = E(I) + ETHETA(I)
            ES(I) = E(I) + ES(I)
13    CONTINUE

*
* Generate the time history for TH seconds with an
* time interval of 0.25 seconds
*

        NTIME = 4.0*NINT(TH)
        DO 14 I = 1, NTIME+1
            T = FLOAT((I-1)/4.0)
            SUM = 0.0
            SUM1 = 0.0
            SUM2 = 0.0
            SUM3 = 0.0
            DO 15 J = 1,NPOINTS
                SUM = ZETA(J)*COS(WE(J)*T + E(J)) + SUM
                SUM1 = HEAV(J)*COS(WE(J)*T + EZ(J)) + SUM1
                SUM2 = PITC(J)*COS(WE(J)*T + ETHETA(J)) + SUM2
                SUM3 = TOT(J)*COS(WE(J)*T + ES(J)) + SUM3
15    CONTINUE
            XG(I) = SUM
            Z(I) = SUM1
            THETA(I) = SUM2
            FP(I) = SUM3
*
* Calculate the total motion at the FP from the assumed time
* histories

        C(I) = Z(I) - 64.20*THETA(I)*PI/180.0
*
* Output the time histories
*

        WRITE(8,1001) T,C(I),FP(I),Z(I),THETA(I),XG(I)
14    CONTINUE
1001  FORMAT(F8.4,5F12.6)

```

```

*
* Calculate the spectrum of the generated time history to check
*
* First calculate the spectral ordinates
*
      DO 200 I = 1,NPOINTS
        CALL FOURIER(Z,I)
        HEAV(I) = ANS
        CALL FOURIER(THETA,I)
        PITC(I) = ANS
        CALL FOURIER(FP,I)
        TOT(I) = ANS
        CALL FOURIER(C,I)
        SC(I) = ANS
        CALL FOURIER(XG,I)
        S(I) = ANS
*
* Output the spectra
*
      WRITE(6,1000) WE(I),HEAV(I),PITC(I),TOT(I),S(I),SE(I)
200    CONTINUE
1000   FORMAT(6F12.5)
      STOP
      END

      SUBROUTINE FOURIER(ZZ,K)
      REAL ZZ(3200),WE(50),TIME(3200),XGC(3200),XGS(3200),DWE(50)
      COMMON/DATA/NTIME,WE,TIME,DWE,ANS
      SUM1 = 0.0
      SUM2 = 0.0
      DO 210 J =1,NTIME
        TIME(J) = FLOAT(J-1)/4.0
        XGS(J) = ZZ(J)*SIN(WE(K)*TIME(J))
        SUM1 = SUM1 + XGS(J)
        XGC(J) = ZZ(J)*COS(WE(K)*TIME(J))
        SUM2 = SUM2 + XGC(J)
210    CONTINUE
      A = SUM2*2.0/FLOAT(NTIME)
      B = SUM1*2.0/FLOAT(NTIME)
*
* Calculate the spectral ordinates
*
      ANS = (A*A+B*B)/2.0/DWE(K)
      RETURN
      END

```








Figure 1


SPECTRA



Encountered Wave



Heave



Pitch

Figure 2

TIME HISTORIES




Figure 3

VARIATION OF HEAVE-PITCH CORRELATION COEFFICIENT WITH HEADING




Figure 4

TOTAL MOTION SPECTRA




Figure 5

TOTAL MOTION TIME HISTORY

REPORT DOCUMENTATION PAGE

DRIC Reference number (if known)

Overall security classification of sheet **UNLIMITED**

(As far as possible this sheet should contain only unclassified information. If it is necessary to enter classified information, the field concerned must be marked to indicate the classification, eg (R), (C) or (S).

Originator's Reference/Report No ARE TM(UHR)90313		Month NOVEMBER	Year 1990
Originator's Name and Location ADMIRALTY RESEARCH ESTABLISHMENT HASLAR GOSPORT HANTS PO12 2AG			
Monitoring Agency Name and Location			
Title GENERATING SHIP MOTION TIME HISTORIES			
Report Security Classification UNLIMITED		Title Classification (U, R, C or S)	
Foreign Language Title (In the case of translations)			
Conference Details			
Agency Reference		Contract Number and Period	
Project Number		Other References	
Authors P CROSSLAND			Pagination and Ref
<p>Abstract</p> <p>PAT-86 is a suite of seakeeping computer programs which may be used for predicting motion spectra of ships in waves. This Technical Memorandum describes a method of generating time histories from these motion spectra. A computer program which generates time histories has been written and is listed in the Appendix.</p> <p>Comparisons of the motion spectra from PAT-86 are made with the spectra of the generated time histories. These comparison are shown to be satisfactory.</p> <div>Abstract Classification (U, R, C or S)</div>			
Descriptors			
Distribution Statement (Enter any limitations on the distribution of the document) UNLIMITED			